
ABSTRACT 

The High Temperature Gas-Cooled Reactor (HTGR) Cost Model was 
developed at the Idaho National Laboratory for the Next Generation Nuclear 
Plant Project.  The HTGR Cost Model calculates an estimate of the capital costs, 
annual operating and maintenance costs, and decommissioning costs for an 
HTGR.  The user can generate these costs for multiple reactor outlet 
temperatures; with and without power cycles, including either a Brayton or 
Rankine cycle; for the demonstration plant, first of a kind, or nth of a kind project 
phases; for a single or four-pack configuration; and for a reactor size of 350 or 
600 MWt. 

This users’ manual contains the mathematical models and operating 
instructions for the HTGR Cost Model.  Instructions, screenshots, and examples 
are provided to guide the user through the HTGR Cost Model.  This model was 
designed for users who are familiar with the HTGR design and Excel.  
Modification of the HTGR Cost Model should only be performed by users 
familiar with Excel and Visual Basic. 

HTGR Cost Model 
Users’ Manual 

1. INTRODUCTION 

The Next Generation Nuclear Plant (NGNP) Project is being conducted under U.S. Department of 
Energy (DOE) direction to meet a national strategic need identified in the 2005 Energy Policy Act to 
promote reliance on safe, clean, economic nuclear energy and to establish a greenhouse-gas-free 
technology for the production of hydrogen. The NGNP represents an integration of high-temperature 
reactor technology with advanced hydrogen, electricity, and process heat production capabilities, thereby 
meeting the mission need identified by DOE. The strategic goal of the NGNP Project is to broaden the 
environmental and economic benefits of nuclear energy in the U.S. economy by demonstrating its 
applicability to market sectors not being served by light water reactors. A high temperature gas-cooled 
reactor (HTGR) differs from a third generation light water reactor by using helium instead of water as the 
coolant, graphite instead of water as the moderator, and tristructural isotropic fuel instead of metal-clad 
fuel. With these features, an HTGR is capable of operating at higher temperatures, which offers a broader 
application to industrial processes and higher thermal efficiencies than are achievable with the lower 
operating temperatures of light water reactors. 

The HTGR can be applied in many industrial applications as a substitute for burning of fossil fuels, 
including natural gas and coal, in addition to producing electricity.  An objective of the NGNP Project is 
to evaluate the economics of heat and power production from an HTGR for incorporation into a variety of 
chemical processes, as well as for more traditional electricity production.  However, in order to properly 
assess the economics of HTGR integration, it was first necessary to determine the capital costs, operating 
and maintenance (O&M) costs, and decommissioning costs of the HTGR.  This was accomplished 
through the creation of the HTGR Cost Model which was used to generate the costs documented in 
TEV-1196, Assessment of High Temperature Gas-Cooled Reactor (HTGR) Capital and Operating Costs 
(Idaho National Laboratory [INL] 2011). 

This users’ manual contains the mathematical models and operating instructions for the HTGR Cost 
Model.  Instructions, screenshots, and examples are provided to guide the user through the HTGR Cost 
Model.  This model was designed for users who are familiar with the HTGR design and Excel.  
Modification of the HTGR Cost Model should only be performed by users familiar with Excel and Visual 
Basic.  The bases and assumptions for the cost model are covered in TEV-1196 and will not be discussed 
in the associated users’ manual; rather the users’ manual will only discuss the model methodology. 

The cost model was developed to estimate HTGR costs for a range of reactor outlet temperatures 
(ROT), 750°C to 950°C, in 50°C increments; with and without power cycles, including both Rankine and 
Brayton cycles; for single and four-pack reactor configurations; for both 350 MWt and 600 MWt power 
levels; for the NGNP, first of a kind (FOAK), and nth of a kind (NOAK) project phases.  A four-pack 
reactor plant includes four reactor vessels with shared control facilities.  The NGNP plant is the initial 
demonstration of the current HTGR design, a FOAK plant is the first commercial installation of the 
HTGR technology, while the NOAK plant is a mature commercial installation of the HTGR technology.  
O&M costs were developed from reactor vendor labor estimates and INL labor estimates, as described in 
TEV-1196 (INL 2011). 

Capital cost estimates are based on estimates provided by Dominion Engineering, Inc. for the INL 
through a separate subcontract.  This users’ manual does not address Dominion’s estimate, which is 
presented and discussed in TEV-1196 (INL 2011).  The Association for the Advancement of Cost 
Engineering (AACE) International recognizes five classes of estimates.  The level of project definition for 

this study was determined to be an AACE International Class 4 estimate, which has a probable error of 
-30% and +50%, as described in TEV-1196 (INL 2011).  A Class 4 estimate is associated with a 
feasibility study or top-down cost estimate and has one to fifteen percent of full project definition 
(AACE 2005). 

2. QUICK USERS’ GUIDE 

The HTGR Cost Model (HTGR Cost Model.xlsm) can be used to quickly generate the capital, O&M, 
fuel, and decommissioning costs for a variety of HTGR configurations in 2009 dollars.  After the Excel 
file is opened, the user is presented with the HTGR Cost Summary worksheet; next the user selects the 
desired configuration from the combo boxes labeled Reactor Inputs (cells D2 to D8); and the 
corresponding capital, O&M, fuel, and decommissioning costs are presented for the selected HTGR 
configuration (cells F12 to F41).  All costs are calculated as described in TEV-1196 (INL 2011). 

Table 1 outlines the reactor inputs available in the combo boxes and Figure 1 presents a screenshot of 
the HTGR Cost Summary worksheet for four-600 MWt 850°C ROT HTGRs with Rankine Power Cycles, 
for the NOAK project phase, and the vendor staffing plan.  Advanced users can unhide the other 
worksheets used in the model, by clicking the Unhide Sheets button, if they wish to update, review, or 
change the underlying data.  The Hide Sheet button hides the detailed worksheets. The detailed worksheet 
descriptions are presented in Section 3. 

Table 1.  HTGR Cost Model user inputs.  

Reactor 
Size (MWt) 

ROT (°C) 
Project 
Phase 

Number of 
Reactors 

Power Cycle 
Present 

Power 
Cycle Type 

Staffing 
Plan 

350 750 NGNP 1 Yes Rankine Vendor 
600 800 FOAK 4 No Brayton INL Model 

850 NOAK 
900 

  950           

Figure 1.  HTGR Cost Model, quick users’ guide screenshot. 

3. DETAILED WORKSHEET DESCRIPTIONS 

The HTGR Cost Model (HTGR Cost Model.xlsm) is implemented in an Excel spreadsheet with 
Visual Basic Macros enabled; the model was developed in Excel 2007 for Windows operating systems 
(Excel 2007).  However, the model has been successfully operated in other versions of Excel on various 
operating systems.  Still, this users’ manual assumes the user is running Excel 2007 for Windows and all 
screenshots are from this version.  The following worksheets are included in the model: 

 HTGR Cost Summary 

 List Info 

 HTGR Capital Cost 

 O&Ms 

 Yearly Fuel Cost 

 Decommissioning 

 Full Results 

 Correlations 

To ensure Macros are enabled click the ‘Office Button’ and select ‘Excel Options’.  Click the ‘Trust 
Center’ tab on the far left of the ‘Excel Options’ window, and click the ‘Trust Center Settings’ button.  
On the far left of the ‘Trust Center Settings’ window click ‘Macro Settings’ and ensure ‘Enable all 
macros (not recommended; potentially dangerous code can run)’ is selected.  To edit Macros ensure that 
the ‘Show Developer’ tab in the ‘Ribbon’ is checked in the ‘Popular’ tab of the ‘Excel Options’ window, 
next click the ‘Developer’ tab in the ‘Ribbon’ and click the ‘Macros’ button.  Select the Macro you wish 
to edit and click ‘Edit’. 

The following sections are arranged such that the approach for each worksheet is summarized, 
followed by a section which defines any named cells or ranges, a section explaining any special 
formatting, a section defining equations and IF statements contained in the worksheet, and finally a 
section listing Macros used in the worksheet.  Special formatting may include cells highlighted a certain 
color identifying user inputs, etc.  Equations and IF statements defined do not include every equation or 
function used in the workbook, rather complex equations are only specified, i.e. functions which use 
simple sums and averages are not necessarily defined, unless they include an IF statement or lookup 
function. 

All screenshots are presented for an NOAK, four-pack of 600 MWt HTGRs with Rankine power 
cycles, for an ROT of 850°C, and the vendor staffing plan. 

3.1 HTGR Cost Summary Worksheet 

The HTGR Cost Summary worksheet is arranged such that the user selects the reactor inputs from 
several combo boxes.  Based on these selections, the capital costs, yearly O&M costs, fuel costs, and 
decommissioning costs calculated in the workbook are summarized in the HTGR Cost Summary 
worksheet.  Costs are reported in millions of U.S. dollars per the year specified by the user on the List 
Info worksheet, the default year is 2009.  From the combo boxes the user specifies the following inputs: 

 Reactor size 

 ROT 

 Project phase 

 Number of reactors 

 Power cycle present 

 Power cycle type 

 Staffing plan 

After the selections are made, the capital cost summary is populated based on outputs from the HTGR 
Capital Cost worksheet, the yearly O&M cost summary is populated based on outputs from the O&Ms 
worksheet, the fuel cost summary is populated based on outputs from the Yearly Fuel Costs worksheet, 
and the decommissioning cost summary is populated based on outputs from the Decommissioning 
worksheet.  A screenshot of the HTGR Cost Summary sheet is presented in Figure 2. 

Advanced users can unhide the worksheets used in the model, by clicking the Unhide Sheets button, 
if they wish to update, review, or change the underlying data.  The Hide Sheets button hides the detailed 
worksheets.  The Unhide Sheets button is linked to the UnHideSheets Macro and the Hide Sheets button 
is linked to the HideSheets Macro. 

3.1.1 Named Cells/Ranges 

ResultsSummary refers to the cell range F12:F41. 

3.1.2 Worksheet Formatting 

No special formatting is used on the HTGR Cost Summary worksheet. 

3.1.3 Worksheet Equations/IF Statements 

The values in the combo boxes are populated based on the data contained in the List Info worksheet.  
To edit or add to the data in the combo boxes the user must right-click the combo box to be edited and 
select ‘Format Control’.  Next, select the ‘Control’ tab, where the ‘Input range’ and ‘Cell link’ cell 
references are shown.  Select the corresponding named cell range to be included from the List Info 
worksheet for the input range and the appropriate cell link on the List Info worksheet. 

Figure 2.  HTGR Cost Summary worksheet screenshot. 

3.1.4 Worksheet Macros 

The UnHideSheets Macro unhides the hidden sheets in the workbook when the user clicks the Unhide 
Sheets button, which by default includes the following worksheets: 

 List Info 

 HTGR Capital Cost 

 O&Ms 

 Yearly Fuel Cost 

 Decommissioning 

 Full Results 

 Correlations 

The UnHideSheets Macro is presented in Figure 3.  The HideSheets Macro hides all sheets in the 
workbook, except the HTGR Cost Summary sheet, when the user clicks the Hide Sheets button.  The 
HideSheets Macro is show in Figure 4. 

 
Figure 3.  UnHideSheets Macro. 

Figure 4.  HideSheets Macro. 

3.2 List Info Worksheet 

The List Info worksheet provides the data which populates the combo boxes on the HTGR Cost 
Summary worksheet.  Based on the value selected in the combo boxes on the HTGR Cost Summary 
worksheet, the appropriate cell link value is written to the cell for the corresponding reactor variable 
(row 8).  Based on the cell link value, the value selected in the combo box is written to the cell for the 
corresponding reactor variable (row 10).  The following table provides the user inputs and possible values 
used in the HTGR Cost Model: 

Table 2.  HTGR Cost Model user inputs.  

Reactor 
Size (MWt) 

ROT (°C) 
Project 
Phase 

Number of 
Reactors 

Power Cycle 
Present 

Power 
Cycle Type 

Staffing 
Plan 

350 750 NGNP 1 Yes Rankine Vendor 
600 800 FOAK 4 No Brayton INL Model 

850 NOAK 
900 

  950           

 

The analysis year, inflation rate, and analysis year Chemical Engineering Plant Cost Index (CEPCI) 
are also included on the List Info worksheet.  All costs are scaled based on 2009 dollars, which has a 
corresponding CEPCI of 521.9.  To adjust capital costs from 2009 dollars to a later year, the capital costs 

are multiplied by the ratio of the CEPCIs.  For instance, the CEPCI of 2010 is 550.8; therefore the 2009 
base capital costs would be multiplied by 550.8/521.9 to adjust capital costs to 2010 dollars.  An inflation 
rate of 3% is assumed to adjust O&M, fuel, and decommissioning costs to years other than 2009.  To 
adjust costs from 2009 dollars the following equation is used: 

  2009 1 3%   

Figure 5 presents a screenshot of the List Info worksheet. 

 
Figure 5.  List Info worksheet screenshot. 

3.2.1 Named Cells/Ranges 

The following cells names have been assigned on the List Info worksheet: 

 RSize – cell A10 

 ROTSel – cell B10 

 RPhase – cell C10 

 Number – cell D10 

 PCyc – cell E10 

 CycChoice – cell F10 

 Staff – cell G10 

 CEPCIYEAR – cell C12 

 IRT – cell C13 

 CEPCI – cell C14 

The following named ranges have been assigned on the List Info worksheet (used in the combo box 
input range on the HTGR Cost Summary worksheet): 

 Reactor Size – range A2:A3 

 ROT – range B2:B6 

 Phase – range C2:C4 

 ReactorQty – range D2:D3 

 HeatorPower – range E2:E3 

 PowerCycle – range F2:F3 

 StaffPlan – range G2:G3 

 RVariables – range A10:G10 

3.2.2 Worksheet Formatting 

Cells highlighted purple indicate user inputs, these include the cell for Analysis Year (C12), Inflation 
Rate (C13), and Analysis Year CEPCI (C14).  Data validation limitations have been entered for the 
analysis year, such that the user must specify a whole number greater than or equal to 2009.  Similarly, 
data validation limitations for the inflation rate have been entered such that the user must input a value 
between 1% and 5%.  Cells highlighted pink include IF statements, other Excel functions, or equations 
dependent upon user inputs. 

To access the data validation information for a given cell, select the cell in question, select the ‘Data’ 
tab in the ‘Ribbon’, and click the ‘Data Validation’ button, next select the ‘Data Validation’ option from 
the drop down list.  This brings up the ‘Data Validation’ window which includes the ‘Validation criteria’ 
in the ‘Settings’ tab. 

3.2.3 Worksheet Equations/IF Statements 

The following equations and IF statements are included in the List Info worksheet: 

 D3=IF(RPhase="NGNP",1,4) – This equation sets the available number of reactors to 1 when 
the NGNP project phase is selected, as it was assumed the demonstration plant would only 
include a single reactor. 

 F2=IF(PCyc=”Yes”,”Rankine”,” “) – This equation sets the power cycle type to Rankine 
when the power cycle present combo box is set to Yes.   

 F3=IF(PCyc=”Yes”,”Brayton”,” “) – This equation sets the power cycle type to Brayton 
when the power cycle present combo box is set to Yes. 

 A10 to G10=INDEX(array, index) – This equation sets the value selected in row 10 for the 
named ranges and the corresponding combo box index. 

3.2.4 Worksheet Macros 

No Macros are included in the List Info worksheet. 

3.3 HTGR Capital Cost Worksheet 

The HTGR Capital Cost worksheet calculates the total capital cost for the selected HTGR plant 
configuration, as well as includes the data used to calculate the capital costs.  The results of the 
calculations are passed to the HTGR Cost Summary worksheet, see Section 3.1.  Capital costs are the sum 
of preconstruction costs, direct costs, indirect costs, and project contingency.  For a description of the 
references used to tabulate the data included in this worksheet refer to TEV-1196 (INL 2011).  All costs 
are calculated in 2009 dollars.  However, the model is developed such that the capital costs can be scaled 
from 2009 dollars using the CEPCI, as discussed in Section 3.2.  Only the scaled values are passed to the 
HTGR Cost Summary worksheet.  In the examples provided, the analysis year is assumed to be 2009; 
therefore, the scaled values are the same as the calculated values, i.e. 2009 dollars. 

At the top of the HTGR Capital Cost worksheet the user inputs specified in the combo boxes on the 
HTGR Cost Summary worksheet are listed, as shown in Figure 6.   

 
Figure 6.  Reactor input summary on the HTGR Capital Cost worksheet, screenshot. 

3.3.1 Preconstruction Costs 

Preconstruction costs are the sum of the land and land rights costs and licensing and application costs.  
A single reactor is assumed to occupy 50 acres of land, while a four-pack is assumed to occupy 100 acres, 
per TEV-1196 (INL 2011).  Licensing and application costs are specified for the NGNP, FOAK, and 
NOAK project phases for single and four-pack configurations.  Preconstruction costs were assumed to be 
consistent for both 350 MWt and 600 MWt reactor sizes (INL 2011).  Figure 7 presents a screenshot of 
the preconstruction costs section of the HTGR Capital Cost worksheet.  Conditional formatting is used to 
highlight which licensing and application cost should apply to the selected configuration, i.e. the cell is 
highlighted green and corresponds to the value in cell I24. 

Figure 7.  Preconstruction costs on the HTGR Capital Cost worksheet, screenshot. 

3.3.2 Direct Costs 

Direct costs are the sum of the individual estimates for the top capital cost items and the balance of equipment adder, assumed to be 25% of 
the sum of the individual estimates (INL 2011).  The individual estimates are based on costs provided by Dominion Engineering, Inc. through a 
separate subcontract as documented in TEV-1196; cells highlighted in blue are scaled as described in the TEV (INL 2011).  Figure 8 presents a 
screenshot of the direct costs section of the HTGR Capital Cost worksheet.  Conditional formatting is used to highlight the sum of the individual 
estimates which should apply to the selected configuration, i.e. the cell is highlighted green, AE63, and corresponds to the value shown in cell I65.  
Conditional formatting is also used to hide the costs not associated with the selected power cycle, i.e. a Rankine cycle is selected for the example 
shown; hence, the rows corresponding to costs for the Brayton cycle are blank (rows 46, 47, 60, and 61). 

 
Figure 8.  Direct costs on the HTGR Capital Cost worksheet, screenshot. 

3.3.3 Indirect Costs and Total Overnight Capital Costs 

Indirect costs are the sum of design costs, construction services, home office and engineering 
services, field office and engineering services, and owner’s costs.  Design costs are the sum of conceptual 
design, preliminary design, final design, and research and development costs.  The majority of design 
costs occur only for the NGNP project phase, only a fraction of the final design costs occur for the FOAK 
and NOAK project phases (INL 2011).  Construction, home office and engineering services, field office 
and engineering services, and owner’s costs are all assumed to be a fraction of the total direct costs: 20%, 
16%, 10%, and 12%, respectively (INL 2011).  The base construction costs are the sum of 
preconstruction, direct, and indirect costs.  A project contingency of 20% is assumed for all project phases 
(INL 2011).  The total overnight cost is the sum of the base construction costs and the project 
contingency.  Figure 9 presents a screenshot of the indirect costs and total overnight capital costs sections 
of the HTGR Capital Cost worksheet. 

3.3.4 Named Cells/Ranges 

No named cells or ranges are included in the HTGR Capital Cost worksheet. 

3.3.5 Worksheet Formatting 

Cells highlighted blue indicate individual cost estimate data that was scaled from other known data 
values.  Cells highlighted pink include IF statements, other Excel functions, or equations dependent upon 
user inputs.  Conditional formatting is used in row 22 which highlights the selected configuration 
licensing costs in green.  Similarly, the selected configuration for the direct costs individual estimates 
(rows 29 and 63) is highlighted green according to the inputs specified by the user.  Finally, conditional 
formatting is also used to hide the costs not associated with the selected power cycle configurations in the 
individual cost estimates (rows 46, 47, 48, 60, 61, and 62).  To identify the conditional formatting used in 
any sheet, select the ‘Home’ tab on the ‘Ribbon’, select ‘Conditional Formatting’, and select ‘Manage 
Rules’ from the drop down list; this will list all conditional formatting on the given worksheet.   

Figure 9.  Indirect and total overnight costs on the HTGR Capital Cost worksheet, screenshot. 

3.3.6 Worksheet Equations/IF Statements 

In the preconstruction costs section of the HTGR Capital Cost worksheet, the following equation is 
used to set the acreage for the selected reactor configuration: 

G11=IF(Number=1,50,IF(Number=4,100,)) 

Based on the above equation, if the user selects a single reactor cell G11 will display 50 acres, and if the 
user selects a four-pack configuration the cell will display 100 acres.  Licensing and application costs are 

also presented for multiple reactor configurations, as a result, an IF statement is used in cell I24 which 
selects the appropriate value from row 22 as follows: 

I24=IF(RPhase="NGNP",I22,IF(RPhase="FOAK",IF(Number=1,J22,IF(Number=4,K22,)), 
IF(RPhase="NOAK",IF(Number=1,L22,IF(Number=4,M22,))))) 

The above function sets the selected configuration licensing costs to the appropriate licensing and 
application costs based on the values selected by the user on the HTGR Cost Summary worksheet. 

In the direct costs, individual estimates section, only FOAK cost data for a single reactor was 
provided in the Dominion estimate; as a result the FOAK four-pack data was scaled based on the 
relationship between the NOAK single and four-pack configurations (INL 2011).  The following example 
is provided for calculating the reactor vessel cost for a four-pack of 600 MWt HTGRs, for a FOAK 
project phase, for an ROT of 850°C (cell T52).  The following Excel function is entered: 

T52=AD52/Y52*O52 

where cell AD52 is the reactor vessel cost for an NOAK four-pack 600 MWt configuration at 850°C, cell 
Y52 is the reactor vessel cost for an NOAK single 600 MWt HTGR at 850°C, and cell O52 is the reactor 
vessel cost for an FOAK single 600 MWt HTGR at 850°C.  Similar equations can be found for all FOAK 
four-pack costs for both 600 MWt and 350 MWt modules.   

Item costs were also scaled to determine the costs for the NGNP project phase for all ROTs except 
950°C, which was provided in the estimate from Dominion.  Costs for the other ROTs were scaled based 
the average scaling factor for a single reactor for the FOAK and NOAK phases.  For example, the reactor 
vessel cost for a single 600 MWt HTGR at 850°C ROT was calculated as follows: 

K52=AVERAGE(P52/Q52,Z52/AA52)*L52 

This equation multiples the scaled 900°C ROT NGNP cost by the average ratios for the 900°C and 850°C 
ROTs for the single FOAK and NOAK project phases. 

The total costs for the individual estimates for top capital cost items are summed based on the power 
cycle selected by the user on the HTGR Cost Summary sheet.  The following function is entered into cell 
AE63, the total cost for a NOAK, 600 MWt four-pack, for an ROT of 850°C: 

AE63=SUM(AE51:AE59,IF(PCyc="yes",IF(CycChoice="Brayton",AE60:AE61,IF(CycChoice= 
"Rankine",AE62,))),) 

This equation sums the individual item costs, excluding the power cycle components, and then depending 
upon if a power cycle is indicated by the user, the corresponding cycle costs, either Brayton or Rankine, 
selected by the user are added.  The cycle selection also corresponds to the conditional formatting 
described in the section above, i.e. the example shown a Rankine cycle is selected; hence, the Brayton 
cycle parameters are hidden and not included in the summation.  

Since the individual item direct costs are presented for multiple reactor configurations a lookup 
function is used in cell I65 which selects the appropriate value from either row 49 or 63: 

I65=HLOOKUP(ROTSel,IF(RPhase="NGNP",I36:M63,IF(RPhase="FOAK",IF(Number=1,N36:
R63,IF(Number=4,S36:W63,)),IF(RPhase="NOAK",IF(Number=1,X36:AB63,IF(Number=4,
AC36:AG63,))))),IF(RSize=350,14,IF(RSize=600,28,)),TRUE) 

The above function uses the built in HLOOKUP function in Excel (horizontal lookup, i.e. looking across 
rows in an array) to select the appropriate value corresponding to the selected ROT; corresponding to the 
array for a single NGNP configuration, a single or four-pack FOAK configuration, or single or four-pack 

NOAK configuration; from either row 49 or 63 depending upon the reactor size selected [IF(RSize=350, 
14,IF(RSize=600,28,)), tells the function to select a value from array row 14 or 28).  In the example 
provided, the lookup function selects the value in cell AE63 based on the function above and the user 
inputs selected on the HTGR Cost summary worksheet, as the selected configuration direct cost total 
shown in cell I65. 

IF statements used in the indirect cost section of the HTGR Capital Cost worksheet are limited to 
calculation of design costs, based on the selected project phase.  The following equations are used for the 
corresponding design costs: 

 Conceptual design – I77=IF(RPhase="NGNP",84,0) 

 Preliminary design – I78=IF(RPhase="NGNP",182,0) 

 Final design – I79=IF(RPhase="NGNP",296,20) 

 R&D – I80=IF(RPhase="NGNP",452,0) 

These equations set the design cost values for the NGNP project phase to the values described in 
TEV-1196, when the user sets the project phase to NGNP, otherwise all design costs are zero, except final 
design which has a cost of 20 (INL 2011). 

The total overnight cost, presented in millions of dollars (cells I106 and I109), is converted to dollars 
per kWt using the following equation (only cell I106 is shown): 

I106=I105*1000000/(RSize*1000*Number) 

3.3.7 Worksheet Macros 

No Macros are included in the HTGR Capital Cost worksheet. 

3.4 O&Ms Worksheet 

The O&Ms worksheet calculates O&M costs for the selected HTGR plant configuration.  The results 
of the calculations are passed to the HTGR Cost Summary worksheet, see Section 3.1.  O&M costs are 
the sum of payroll costs, various industry fees, insurance and taxes, material supplies services and 
upgrades, outage costs, and administration and general cost overhead (INL 2011).  The worksheet 
includes the INL staffing estimate for a single and additional unit for both 600 MWt and 350 MWt plants, 
as outlined in TEV-1196 (INL 2011).  A screenshot for the staffing plan is not included, as it identifies 
close to 200 positions and is prohibitively long, for the detailed staffing plans see TEV-1196 or the 
O&Ms worksheet.  Only the total plant staffing level for the INL analysis is presented in the screenshot 
for the O&Ms worksheet.  Additionally, the staffing levels identified in vendor estimates are presented; 
however, the breakdown by position is not included for the vendor estimate.  Figure 10 presents a 
screenshot of the O&Ms worksheet.  All O&M costs are presented in 2009 dollars; the inflation rate is 
used to adjust O&M costs to years other than 2009, as outlined in Section 3.2. 

Figure 10.  O&Ms worksheet screenshot. 

3.4.1 Named Cells/Ranges 

No named cells or ranges are included in the O&Ms worksheet. 

3.4.2 Worksheet Formatting 

Cells highlighted pink include IF statements, other Excel functions, or equations dependent upon user 
inputs.  Conditional formatting is used in rows 245 and 246, which highlights the costs used in the O&M 
calculations in green.  To identify the conditional formatting used in any sheet, select the ‘Home’ tab on 
the ‘Ribbon’, select ‘Conditional Formatting’, and select ‘Manage Rules’ from the drop down list; this 
will list all conditional formatting on the given worksheet.   

3.4.3 Worksheet Equations/IF Statements 

In the O&Ms worksheet, the following equation is used to set the staffing estimate based on the user 
selection in the HTGR Cost Summary worksheet, cell F229 is used as the example: 

F229=IF(Staff="Vendor",F226,IF(Staff="INL Model",F225,)) 

This equation sets the cell value (row 229) to either the INL or vendor staffing estimate (row 225 or 226, 
respectively) based on the value selected by the user.  The selected O&M cost is calculated by the 
following equation in cell F248: 

F248=(IF(RSize=600,F246,IF(RSize=350,H246,))+IF(Number=4,IF(RSize=600,(3*G246), 
IF(RSize=350,(3*I246),))))/1000000 

This equation selects either the 600 MWt or 350 MWt costs for the single reactor, then adds three 
times the appropriate additional unit costs if a four-pack is selected.  Finally, the O&M cost per MWt-hr 
is calculated using the following equation: 

F249=F248*1000000/(RSize*Number)/(365*24) 

It is assumed that O&M costs are consistent for all project phases (INL 2011). 

3.4.4 Worksheet Macros 

No Macros are included in the O&Ms worksheet. 

3.5 Yearly Fuel Cost Worksheet 

The Yearly Fuel Cost worksheet calculates fuel costs for the selected HTGR plant configuration.  The 
results of the calculations are passed to the HTGR Cost Summary worksheet, see Section 3.1.  Fuel costs 
are the sum of uranium ore, uranium conversion, uranium enrichment, tails disposal, fuel fabrication, 
on-site spent fuel storage, and spent fuel disposition costs (INL 2011).  The number of fuel blocks per 
reload is dependent upon the size and number of reactors in the selected configuration, as is the kilograms 
of heavy metals in the refueling blocks (INL 2011).  The calculations for the amount of uranium, U3O8, 
and separative work units (SWU) are detailed in TEV-1196; however, they will also be summarized in 
this document (INL 2011).  Figure 11 presents a screenshot of the Yearly Fuel Cost worksheet.  All fuel 
costs are presented in 2009 dollars; the inflation rate is used to adjust fuel costs to years other than 2009, 
as outlined in Section 3.2. 

3.5.1 Named Cells/Ranges 

No named cells or ranges are included in the Yearly Fuel Cost worksheet. 

3.5.2 Worksheet Formatting 

Cells highlighted pink include IF statements, other Excel functions, or equations dependent upon user 
inputs. 

Figure 11.  Yearly Fuel Cost worksheet screenshot. 

3.5.3 Worksheet Equations/IF Statements 

Data and specific references for the equations described in this section are presented in TEV-1196; 
please refer to that document for detailed information on the following calculations (INL 2011).  In the 
Yearly Fuel Cost worksheet, the following equation is used to calculate the number of fuel blocks in each 
fuel reloading: 

B2=510*RSize/600*Number 

The mass of heavy metals, in the refueling blocks, is calculated as follows: 

B9=(1748+514)*RSize/600*Number 

The mass of uranium fed per year is next calculated based on the following equation (see page 40 of 
INL 2011): 

B11=B10*(B4-B7)/(B8-B7) 

The amount of uranium tailings per year is then calculated (see page 40 of INL 2011): 

B14=B10*(B4-B8)/(B8-B7) 

Finally, the SWUs are calculated (see page 40 of INL 2011): 

B15=B10*(1-2*B4)*LN((1-B4)/B4)+B14*(1-2*B7)*LN((1-B7)/B7)-B11*(1-2*B8)* 
LN((1-B8)/B8) 

The above calculations provide the fuel requirements to calculate the yearly costs that make up the 
total annual refueling costs and the refueling cost per core.  The majority of costs are consistent for all 
project phases, excluding fuel fabrication costs, for which costs are higher for the NGNP and FOAK 
project phases.  The fuel fabrication unit cost is defined as follows: 

B22=IF(OR(RPhase="NGNP",RPhase="FOAK"),25000,IF(RPhase="NOAK",10000,))* 
(1+IRT)^(2009-2007) 

The fuel fabrication cost is based on 2007 values; thus, the costs are adjusted to 2009 values using the 
inflation rate specified in the List Info worksheet.  Yearly costs (column D) are calculated by multiplying 
the unit cost by the annual yearly requirement (column B).  The total cost is the sum of all the yearly 
costs, and the cost per core is calculated as follows: 

B26=D25/Number/B6 

Finally, the fuel cost per MWt-hr is calculated using the following equation: 

B31=D28*1000000/(RSize*Number)/(365*24) 

3.5.4 Worksheet Macros 

No Macros are included in the Yearly Fuel Cost worksheet. 

3.6 Decommissioning Worksheet 

The Decommissioning worksheet calculates decommissioning costs for the selected HTGR plant 
configuration.  The results of the calculations are passed to the HTGR Cost Summary worksheet, see 
Section 3.1.  Decommissioning costs were estimated based on the methodology presented in 
NUREG-1307, Rev. 14.  As this document is for estimation of decommissioning costs for large light 
water reactors (LWRs), the decommissioning costs were converted to a dollar per MWt basis in order to 
estimate the costs for decommissioning the HTGR (INL 2011).  Figure 12 presents a screenshot of the 
Decommissioning worksheet.  All decommissioning costs are presented in 2009 dollars; the inflation rate 
is used to adjust fuel costs to years other than 2009, as outlined in Section 3.2. 

Figure 12.  Decommissioning worksheet screenshot. 

3.6.1 Named Cells/Ranges 

No named cells or ranges are included in the Decommissioning worksheet. 

3.6.2 Worksheet Formatting 

Cells highlighted pink include IF statements, other Excel functions, or equations dependent upon user 
inputs.  

3.6.3 Worksheet Equations/IF Statements 

The decommissioning cost for an average light water reactor, as specified in NUREG-1307, was 
calculated using the following equation (see page 41 of INL 2011): 

D12=D2*(0.65*F6+0.13*D3+0.22*F10) 

The dollar per MWt for decommissioning is calculated based on a 3,400 MWt LWR.  This average cost 
per MWt is used to calculate the total decommissioning cost as follows: 

D17=D16*RSize*Number/1000000 

3.6.4 Worksheet Macros 

No Macros are included in the Decommissioning worksheet. 

3.7 Full Results Worksheet 

The Full Results worksheet tabulates the results summarized on the HTGR Cost Summary worksheet 
for all possible configuration iterations, excluding the staffing plan, which is set equal to the current user 
selection.  To populate the tables, click the Populate Tables for Selected Staffing Plan button.  This button 

is linked to the FullResults Macro.  A partial screenshot of the Full Results worksheet is shown in 
Figure 13.  Note, all reactor sizes and phases are included on this sheet, not just the NGNP project phase 
as shown in the partial screenshot. 

 
Figure 13.  Full Results worksheet, partial screenshot. 

3.7.1 Named Cells/Ranges 

No named cells or ranges are included in the Full Results worksheet. 

3.7.2 Worksheet Formatting 

No special formatting is used on the Full Results worksheet. 

3.7.3 Worksheet Equations/IF Statements 

No equations or IF statements are included in the Full Results worksheet. 

3.7.4 Worksheet Macros 

The FullResults Macro compiles the full results for the user selected staffing plan into several tables, 
when the user clicks the Populate Tables for Selected Staffing Plan button.  The FullResults Macro is 
presented in Figure 14.  Note this Macro is several pages long. 

 
Figure 14.  FullResults Macro. 



 

 25

 
Figure 14.  FullResults Macro. (continued) 

Figure 14.  FullResults Macro. (continued) 

Figure 14.  FullResults Macro. (continued) 

Figure 14.  FullResults Macro. (continued) 

Figure 14.  FullResults Macro. (continued) 

Figure 14.  FullResults Macro. (continued) 

3.8 Correlations Worksheet 

The Correlations worksheet is used to develop relationships for the HTGR and power cycle costs 
based on the ROT, thermal rating, power cycle rating, number of modules, and project phase.  These 
correlations were developed in order to scale the costs for project variations not included in HTGR Cost 
Model.  For example, these correlations can be used to estimate the cost of six HTGR modules at an ROT 
of 775C for an NOAK HTGR plant.  Separate correlations are developed for the HTGR cost without 
power cycles (i.e. for heat production only), the Brayton cycle cost, and the Rankine cycle cost.  The 
correlations are based on the direct costs, less the adders, which can be added to the correlations to 
account for the balance of equipment costs, indirect costs, preconstruction costs, and project contingency 
as described in TEV-1196 (INL 2011).   

The totals for the HTGR less power cycle (rows 4 and14), the Brayton cycle (rows 7 and 17), and the 
Rankine cycle (rows 10 and 20) are determined based on the summation of the corresponding individual 
item costs on the HTGR Capital Cost worksheet.  Based on this information, along with the assumed 
cycle efficiencies (rows 25 and 28), correlations were developed as described in TEV-1196 (INL 2011).  
The correlations are then used to predict the various direct costs (rows 5, 8, 11, 15, 18, 21) and the percent 
error between the model prediction and actual value is also calculated (rows 6, 9, 12, 16, 19, 22).  
Figure 15 presents a screenshot of the Correlations worksheet.  Charts comparing the correlations and 
predicted values are also included on this sheet; however, the screenshot does not include these charts. 

Figure 15.  Correlations worksheet screenshot. 

3.8.1 Named Cells/Ranges 

No named cells or ranges are included in the Correlations worksheet. 

3.8.2 Worksheet Formatting 

Cells highlighted pink include IF statements, other Excel functions, or equations dependent upon user 
inputs. 

3.8.3 Worksheet Equations/IF Statements 

All ROT correlations were developed using the costs for a single 600 MWt unit for an NOAK project 
phase.  The ROT correlation is developed using the SLOPE and INTERCEPT Excel functions.  Both the 
HTGR and Brayton cycle correlations are linear step functions due to materials changed (INL 2011).  The 
following equation is used to calculate the slope for ROTs less than 850°C for the HTGR with no power 
cycle: 

D34=SLOPE(S14:U14,S3:U3) 

where S14:U14 are the known y-values, and S3:U3 are the known x-values.  The following equation is 
used to calculate the intercept: 

D35=INTERCEPT(S14:U14,S3:U3) 

Similar equations are used for the remaining HTGR ROT ranges and for the Brayton and Rankine 
cycle ROT correlations.  An exponential function is used in the correlation for the number of units.  The 
exponent was determined by averaging the exponents for all ROTs for the single and four-pack NOAK 
costs.  The following equation is used to set the exponent for the HTGR with no power cycle: 

H34=AVERAGE(LN(X14/S14)/LN(4),LN(Y14/T14)/LN(4),LN(Z14/U14)/LN(4),LN(AA14/ 
V14)/LN(4),LN(AB14/W14)/LN(4)) 

Again, similar equations are used for the Brayton and Rankine cycles.  The correlation for reactor size 
is also in the form of an exponential function.  The exponent was determined by averaging the exponents 
for all ROTs for the single 600 MWt and 350 MWt NOAK costs, 600 MWt is the basis.  The following 
equation is used to set the exponent for the HTGR with no power cycle: 

J34=AVERAGE(LN(S14/S4)/LN(600/350),LN(T14/T4)/LN(600/350),LN(U14/U4)/LN(600/ 
350),LN(V14/V4)/LN(600/350),LN(W14/W4)/LN(600/350)) 

Similar equations are used for the Brayton and Rankine cycles; however the ratio of the power cycle 
sizes are used (rows 26, 27, 29, and 30) instead of the thermal rating ratio.  It should be noted that the 
gross heat input to the power cycle could also be used in place of the power cycle size.   

Finally, multipliers for the FOAK and NGNP plants were determined by averaging the multipliers for 
each ROT for the single 600 MWt and 350 MWt configurations, with the NOAK plant as the basis, i.e. a 
multiplier of one.  The equation for the FOAK correlation for the HTGR with no power cycle is provided 
as an example: 

L34=AVERAGE(AVERAGE((I14/S14),(J14/T14),(K14/U14),(L14/V14),(M14/W14)), 
AVERAGE((I4/S4),(J4/T4),(K4/U4),(L4/V4),(M4/W4))) 

Both 600 MWt and 350 MWt plants were used to determine the reactor phase multipliers in order to 
minimize the percent error for a variety or reactor sizes.  

The correlations were then used to predict costs for the HTGR, Brayton cycle, and Rankine cycle.  
The calculation for a 350 MWt NGNP HTGR with no power cycle and an ROT of 750°C (cell D5) is 
provided as the example: 

D5=$N$34*((350/600)^$J$34)*($D$2^$H$34)*IF(D$3<=850,$D$34*D$3+$D$35,IF(AND( 
D$3>850,D$3<=900),$E$34*D$3+$E$35,IF(D$3>900,$F$34*D$3+$F$35,))) 

For detailed descriptions of the correlations developed in the HTGR Cost Model see pages 22 through 
29 of TEV-1196 (INL 2011). 

3.8.4 Worksheet Macros 

No Macros are included in the Correlations worksheet. 

